
Session 4: Advanced Menu Programs Continued 
 

At the end of this lesson participants will be able to: 

• Implement automatic synchronization 

• Create dashboards 

• Create rich reports using HTML templates 

Automatic synchronization 
We know how to use both simple synchronization and the synchronization logic functions to implement 

synchronization initiated by the interviewer or supervisor. In some cases, we would prefer to have the 

synchronization with headquarters happen automatically. We can do this using logic by synchronizing 

every time the supervisor comes to the main menu. Since the supervisor will very frequently come to 

the main menu this will force the supervisor to send data often. 

To synchronize every time the supervisor reaches the menu, we will write a new function checkForSync 

that we will call from the onfocus of the supervisor main menu. 

PROC SUPERVISOR_MAIN_MENU 

onfocus 

// Clear previous choice 

$ = notappl; 

 

// Automatically sync 

checkForSync(); 

 

To avoid annoying the supervisor with overly frequent synchronizations we can save the last time we 

synced using savesetting and then only sync if a certain amount of time has passed since the last sync.  

function checkForSync() 

 

    // Time to allow between syncs to prevent too many syncs 

    numeric minutesBetweenSyncs = 3 * 60; // 3 hours     

    numeric lastSyncTime = tonumber(loadsetting("lastSyncTime")); 

    numeric minutesSinceLastSync = (timestamp() - lastSyncTime)/60;  

    if minutesSinceLastSync > minutesBetweenSyncs  

       or lastSyncTime = default then 

        if syncToHeadquarters() then 

                savesetting("lastSyncTime", maketext("%d", timestamp())); 

        endif; 

    endif; 

end; 

 

Here we use the timestamp() function which returns the time in seconds. When computing differences 

between times it is simpler to use timestamp() rather than systime() since you don't have to worry 

about the difference between times on different days. 

We can be a bit more clever and also not sync if the device is not connected to the internet by using the 

connection() function. 

function checkForSync() 

 

    // Time to allow between syncs to prevent too many syncs 

    numeric minutesBetweenSyncs = 3 * 60; // 3 hours 

     

    numeric lastSyncTime = tonumber(loadsetting("lastSyncTime")); 

    numeric minutesSinceLastSync = (timestamp() - lastSyncTime)/60;  

    if minutesSinceLastSync > minutesBetweenSyncs  

       or lastSyncTime = default then 

        // Only sync if there is an internet connection 

        if connection() then 

            if syncToHeadquarters() then 

                savesetting("lastSyncTime", maketext("%d", timestamp())); 

            endif; 

        endif; 

    endif; 

end; 

 

The syncToHeadquarters() uses the usual logic for synchronizing based on syncconnect(), syncdata() and 

syncfile(). 

function syncToHeadquarters() 

    if syncconnect(csweb, "http://csweb.teleyah.com/api") then 

 

         // Upload listing and household data from supervisor to HQ 

         syncdata(PUT,HOUSEHOLD_DICT); 

         syncdata(PUT,LISTING_DICT);  

          

         // Download latest staff file 

         syncdata(GET, STAFF_DICT); 

          

         // Sync the listing assignments both ways 

         syncdata(BOTH, LISTINGASSIGNMENTS_DICT); 

          

         // Get latest programs from HQ 

         syncfile(GET,"PHS/Household/PopstanHouseholdSurvey.pff", 

                      "../Household/"); 

         syncfile(GET,"PHS/Household/PopstanHouseholdSurvey.pen",  

                      "../Household/"); 

         syncfile(GET,"PHS/Listing/Listing.pff",  

                      "../Listing/"); 

         syncfile(GET,"PHS/Listing/Listing.pen", "../Listing/"); 

         syncfile(GET,"PopstanHouseholdSurvey/Menu/Menu.pen"); 

         syncfile(GET,"PopstanHouseholdSurvey/Menu/Menu.pff"); 

                          

         syncdisconnect(); 

 

         syncToHeadquarters = 1; 

    else 

        syncToHeadquarters = 0;  

    endif; 

end; 

Dashboards 
To give the interviewer a snapshot of their work remaining we can create a simple dashboard using the 

question text of the interviewer main menu. We will show the following text: 

Logged in as interviewer %currentLoginName% 
 
Current work: 
 
Listing: %clustersCompletedByInterviewer% completed of %clustersAssignedToInterviewer% 
assigned clusters 
 
Households: %householdsCompletedByInterviewer% completed of  
%householdsAssignedToInterviewer% assigned 

 
We will use the following global variables to store the fills: 

string currentLoginName; 

numeric clustersAssignedToInterviewer; 

numeric clustersCompletedByInterviewer; 

numeric householdsAssignedToInterviewer; 

numeric householdsCompletedByInterviewer; 

We could also use functions, however, counting the number of completed and assigned households 

involves looping through the entire assignments file which could get slow after a lot of households have 

been assigned. Instead we can calculate this information once when the interviewer logs in. The only 

actions that can cause it to change are listing, interviewing and syncing with the supervisor. After listing 

or interviewing the interviewer will pass through the login screen anyway so we only need to update 

these variables during login and after synchronization with the supervisor. 

PROC LOGIN 

preproc 

 

// Check for saved login 

if loadsetting("login") <> "" then 

    LOGIN = tonumber(loadsetting("login")); 

    noinput; 

endif; 

 

postproc 

// verify that staff code exists 

// in lookup file 

if loadcase(STAFF_DICT, LOGIN) = 0 then 

    errmsg("Invalid login. Try again."); 

    reenter; 

else 

    // Loadcase worked so login code is valid 

    currentLoginName = STAFF_NAME; 

 

    // Store login in settings so we can use it 

    // on next login 

    savesetting("login", maketext("%d",LOGIN)); 

     

    if STAFF_ROLE = 1 then 

        // Interviewer 

        updateInterviewerDashboard(); 

        skip to INTERVIEWER_MAIN_MENU; 

    else         

        // Supervisor 

        skip to SUPERVISOR_MAIN_MENU; 

    endif;   

endif; 

 

The function updateInterviewerDashboard will loop through the listing and household assignments files 

and count the number of assigned and completed clusters and households. 

function updateInterviewerDashboard() 

    clustersAssignedToInterviewer = 0; 

    clustersCompletedByInterviewer = 0; 

     

    forcase LISTINGASSIGNMENTS_DICT where LA_ASSIGNED_INTERVIEWER = LOGIN  

                                          and LA_CLUSTER_STATUS = 2 do 

        inc(clustersAssignedToInterviewer); 

        if isClusterComplete(LA_PROVINCE, LA_DISTRICT, LA_CLUSTER) then 

            inc(clustersCompletedByInterviewer); 

        endif; 

    endfor; 

     

    householdsAssignedToInterviewer = 0; 

    householdsCompletedByInterviewer = 0; 

     

    forcase HOUSEHOLDASSIGNMENTS_DICT where HA_ASSIGNED_TO = LOGIN  

                   and getClusterStatus(HA_PROVINCE, HA_DISTRICT,  

                                        HA_CLUSTER) = 3 do 

        inc(householdsAssignedToInterviewer); 

         

        if isHouseholdComplete(HA_PROVINCE, HA_DISTRICT, HA_CLUSTER,   

                               HA_HOUSEHOLD_NUMBER) then 

            inc(householdsCompletedByInterviewer); 

        endif; 

    endfor; 

end; 

 

This function uses the following helper functions to determine the statuses of households and clusters: 

function isClusterComplete(prov, dist, clust) 

    LI_PROVINVE = prov; 

    LI_DISTRICT = dist; 

    LI_CLUSTER = clust; 

    if loadcase(LISTING_DICT, LI_PROVINVE, LI_DISTRICT, LI_CLUSTER) 

        and not ispartial(LISTING_DICT) then 

        isClusterComplete = 1; 

    else 

        isClusterComplete = 0; 

    endif; 

end; 

 

function getClusterStatus(prov, dist, clust) 

    LI_PROVINVE = prov; 

    LI_DISTRICT = dist; 

    LI_CLUSTER = clust; 

    loadcase(LISTINGASSIGNMENTS_DICT, LI_PROVINVE, 

                LI_DISTRICT, LI_CLUSTER); 

    getClusterStatus = LA_CLUSTER_STATUS; 

end; 

function isHouseholdComplete(prov, dist, clust, hhnum) 

    PROVINCE = prov; 

    DISTRICT = dist; 

    CLUSTER = clust; 

    HOUSEHOLD_NUMBER = hhnum; 

    if loadcase(HOUSEHOLD_DICT, PROVINCE, DISTRICT, CLUSTER, 

                HOUSEHOLD_NUMBER) and  

        not ispartial(HOUSEHOLD_DICT) then 

        isHouseholdComplete = 1; 

    else 

        isHouseholdComplete = 0; 

    endif; 

end; 

Group Exercise 

Create a dashboard for the supervisor. Use the question text for the supervisor main menu to show the 

following: 

Logged in as supervisor %currentLoginName% 
 

Current work: 
 

Listing: %clustersCompletedByAllInterviewers% completed of  
%clustersAssignedToAllInterviewers% assigned clusters 

 
Households: %householdsCompletedByAllInterviewers% completed of   
%householdsAssignedToAllInterviewers% assigned 
 

Create the appropriate global variables for the fills in the question text and a function 

updateSupervisorDashboard(). Call your function when the supervisor logs in, after syncing with an 

interviewer and after modifying cluster or household assignments. The function should calculate the 

assigned and interviewed clusters and households the same way that the updateInterviewerDashboard 

does except that it should count households and clusters assigned to all interviewers rather than the 

only to the one logged in. 

 

Reports 
CSPro supports generating rich reports using HTML templates that can be run on mobile or on desktop. 

As an example, lets create the progress report.  

To launch the report from CSPro we need to implement the reports menu from the supervisor menu. 

Add a new item to the menu dictionary named REPORTS with value set Completion – 1, Household 

Status – 2, Back – 9. Add the item to the form and skip to it from the supervisor main menu when the 

supervisor picks reports. In the postproc for REPORTS we will call a function that we will create named 

progressReport(). 

PROC REPORTS 

onfocus 

$ = notappl; 

 

postproc 

 

if $ = 1 then 

    progressReport(); 

    reenter; 

elseif $ = 9 then 

    reenter SUPERVISOR_MAIN_MENU; 

endif; 

The progress report will show the number of households assigned, number complete, number 

remaining and number partially complete. We will display it in a table: 

Assigned Complete Partial Remaining 

    

 

The HTML for the table would look like: 

<table> 

    <thead> 

        <th>Assigned</th> 

        <th>Complete</th> 

        <th>Partial</th> 

        <th>Remaining</th> 

    </thead> 

    <tbody> 

      <tr> 

        <td></td> 

        <td></td> 

        <td></td> 

        <td></td> 

      </tr> 

    </tbody> 

</table> 

 

We need to calculate the numbers of assigned, complete, partial and remaining cases and insert those 

values into the table data (<td>) nodes. To support inserting values into an HTML document, CSPro uses 

templates. A template is an HTML document that contains variables in it that can be replaced by values 

from CSPro logic. CSPro uses a templating system called Mustache, which is used by many websites and 

is well documented online. Any variable to be replaced by CSPro logic must be surrounded by two curly 

brackets e.g. {{variable}}. For our table we would use: 

  

https://mustache.github.io/

<table> 

    <thead> 

        <th>Assigned</th> 

        <th>Complete</th> 

        <th>Partial</th> 

        <th>Remaining</th> 

    </thead> 

    <tbody> 

      <tr> 

        <td>{{assigned}}</td> 

        <td>{{complete}}</td> 

        <td>{{partial}}</td> 

        <td>{{remaining}}</td> 

      </tr> 

    </tbody> 

</table> 

 

In addition, the entire template must be wrapped in a script tag of type "application/vnd.cspro.report-

template" with id "cspro_report_template". This script tag should be part of an HTML document that 

contains an empty div node with id "cspro_report". When the report is displayed, this div node will be 

replaced with the content of the template script tag that has all the variables replaced with values from 

CSPro logic. Using this structure, our HTML file would look like: 

<!doctype html> 

<html lang="en"> 

  <head> 

    <title>Progress Report</title> 

</head> 

<body> 

    <h1>Progress report</h1> 

 

    <div id="cspro_report"><!-- replaced by expanded template --></div> 

 

    <!-- report template --> 

    <script type="application/vnd.cspro.report-template" id="cspro_report_template"> 

        <table> 

        <thead> 

            <th>Assigned</th> 

            <th>Complete</th> 

            <th>Partial</th> 

            <th>Remaining</th> 

        </thead> 

        <tbody> 

            <tr> 

                <td>{{assigned}}</td> 

                <td>{{complete}}</td> 

                <td>{{partial}}</td> 

                <td>{{remaining}}</td> 

            </tr> 

        </tbody> 

    </table> 

    </script> 

</body> 

</html> 

  

To make the page look nicer you can add css and Javascript to the page just as you would for any web 

page. For example, if we add Bootstrap css and Javascript our page it would look like: 

<!doctype html> 

<html lang="en"> 

  <head> 

    <meta charset="utf-8"> 

    <meta name="viewport" content="width=device-width, initial-scale=1, shrink-to-fit=no"> 

    <link rel="stylesheet" href="css/bootstrap.min.css"> 

    <title>Progress Report</title> 

</head> 

<body> 

    <h1>Progress report</h1> 

    <div id="cspro_report"><!-- replaced by expanded template --></div> 

    <!-- report template --> 

    <script type="application/vnd.cspro.report-template" id="cspro_report_template"> 

        <table class="table table-bordered"> 

        <thead> 

            <th>Assigned</th> 

            <th>Complete</th> 

            <th>Partial</th> 

            <th>Remaining</th> 

        </thead> 

        <tbody> 

            <tr> 

                <td>{{assigned}}</td> 

                <td>{{complete}}</td> 

                <td>{{partial}}</td> 

                <td>{{remaining}}</td> 

            </tr> 

        </tbody> 

    </table> 

    </script> 

    <script src="js/jquery-3.3.1.slim.min.js"></script> 

    <script src="js/popper.min.js"></script> 

https://getbootstrap.com/

    <script src="js/bootstrap.min.js"></script> 

</body> 

</html> 

 

Note that if you want your report to work offline, you will need to include css and Javascript that is 

stored on the device that the report is running on. If you have no internet you will not be able to use css 

and Javascript hosted on a CDN or other server. We will download the css and Javascript files we need 

and put them in subfolders of the folder containing our reports named css and js. 

Now that we have our HTML file, we can implement the function progressReport to show the report. 

The first step is to calculate the values for the table cells by looping through the household assignments 

file. 

function progressReport() 

 

    // Compute data for report 

    numeric assignedHouseholds = 0; 

    numeric completeHouseholds = 0; 

    numeric partialHouseholds = 0; 

     

    forcase HOUSEHOLDASSIGNMENTS_DICT where getClusterStatus(HA_PROVINCE,  

                                             HA_DISTRICT, HA_CLUSTER) = 3 do 

     

        // Count every household as assigned 

        inc(assignedHouseholds); 

         

        if loadcase(HOUSEHOLD_DICT, PROVINCE, DISTRICT, CLUSTER,  

                    HOUSEHOLD_NUMBER) then 

            // Households with case in household data file  

            // are either partial or complete 

            if ispartial(HOUSEHOLD_DICT) then 

                inc(partialHouseholds); 

            else 

                inc(completeHouseholds); 

            endif; 

        endif; 

    endfor; 

 

    // Remaining households are assigned households that are not complete 

    numeric remainingHouseholds = assignedHouseholds - completeHouseholds; 

 

The next step is to assign these calculated values to corresponding parameters in the report template. 

This is done using the command setreportdata().  This function takes the name of the parameter, i.e. 

the variable in the HTML file surrounded by {{}}, and the value to replace that variable with. The value 

must be a string so we use maketext() to convert our numeric variables to strings. 

    // Set data to be replaced in report 

    setreportdata("assigned", maketext("%d", assignedHouseholds)); 

    setreportdata("complete", maketext("%d", completeHouseholds)); 

    setreportdata("partial", maketext("%d", partialHouseholds)); 

    setreportdata("remaining", maketext("%d", remainingHouseholds)); 

Finally, we run and display the report with the command report. We pass it the path to the HTML file. If 

we place our HTML file in the Reports subfolder of the menu folder then we would call: 

    // create and view the report 

    report("Reports/progress.html"); 

 

end; 

 

This will display the HTML page with the parameters in the template replaced by the values passed to 

setreportdata(). 

 

Group Exercise 

Modify the table template and the logic in the progressReport function to show the percent complete. 

The percent complete column should be on the far right of the table and should be computed as the 

percentage of complete cases to assigned cases. 

Let's create a better version of the report that breaks out the data for each interviewer. 

 

We could implement this report with 25 parameters in the template – one for each cell in the 5 x 5 

table. However, this will not work if the number of interviewers was to change. We need a way to allow 

the number of rows to be variable depending on the number of interviewers in the staff file. The 

setreportdata command can be passed a repeating record in the dictionary and generate a row in the 

table for each occurrence of the record. Unfortunately, we don't have a repeating record in our 

dictionary for the interviewers. The staff dictionary has a single record for each staff member so we 

can't just pass the staff record to setreportdata. However, we can create a new multiply occurring 

record, copy the data from the staff dictionary to the repeating record and pass the repeating record to 

setreportdata. 

We could create a new record in the menu program for this, however it is common to use a working 

storage dictionary for this purpose. A working storage dictionary is a dictionary that is not associated 

with any file. It contains temporary data that only exists while the application is running. To add a 

working storage dictionary to the application, choose "Add files…" from the files menu and check the 

box next to "Working storage dictionary" at the bottom of the dialog box. 

In the working storage dictionary create a new record named WS_PROGRESS_REC with up to 10 

occurrences. Add items in this record for each column of the report: WS_PROGRESS_INTERVIEWER 

(alpha length 25), WS_PROGRESS_ASSIGNED (numeric length 4), WS_PROGRESS_COMPLETE (numeric 

length 4), WS_PROGRESS_PARTIAL (numeric length 4) and WS_PROGRESS_REMAINING (numeric length 

4). 

In the progessReport() function we will fill in one occurrence of WS_PROGRESS_REC for each 

interviewer. Before we can do that, we need to set the correct number of record occurrences. By 

default, a case in an external or working storage dictionary is created with all of its occurrences i.e. the 

number of max occurrences specified in the dictionary. Using the max occurrences would cause the 

table we generate to have more rows than there are interviewers. To prevent this, we first delete all of 

the default occurrences of the record using the delete command and then we add back just the number 

of occurrences we need using the insert command. 

function progressReport() 

 

    // Compute data for report 

 

    clear(WS_DICT); 

     

    // By default, a working storage dictionary has all occurrences but  

    // that will create too big a table in report so we delete all  

    // occurrences and add back the ones we need 

    do varying numeric i = count(WS_PROGRESS_REC) until i <= 0 by (-1) 

        delete (WS_PROGRESS_REC(i)); 

    enddo; 

         

    numeric i = 1; 

    forcase STAFF_DICT where STAFF_ROLE = 1 do 

        insert(WS_PROGRESS_REC(i)); 

 

Instead of calculating the number of assigned, completed, partial and remaining households once, we 

need to calculate it for each of the interviewers. First, we loop through the staff data file to get each 

interviewer. Inside that loop we need to compute the number of assigned, completed, partial remaining 

cases for just that interviewer. The natural way to do this would be to use a forcase inside another 

forcase, however it is not possible to have nested forcase loops so we need to make one of the loops 

use the old locate/while loadcase technique. 

  



13 
 

    numeric i = 1; 

    forcase STAFF_DICT where STAFF_ROLE = 1 do 

        insert(WS_PROGRESS_REC(i)); 

        WS_PROGRESS_INTERVIEWER(i) = STAFF_NAME; 

        WS_PROGRESS_ASSIGNED(i) = 0; 

        WS_PROGRESS_COMPLETED(i) = 0; 

        WS_PROGRESS_PARTIAL(i) = 0; 

         

        locate(HOUSEHOLDASSIGNMENTS_DICT, >=, ""); 

        while loadcase(HOUSEHOLDASSIGNMENTS_DICT) do 

            if getClusterStatus(HA_PROVINCE, HA_DISTRICT, HA_CLUSTER) = 3  

                and HA_ASSIGNED_TO = STAFF_CODE then 

         

                // Count every household as assigned 

                inc(WS_PROGRESS_ASSIGNED(i)); 

                 

                if loadcase(HOUSEHOLD_DICT, PROVINCE, DISTRICT, CLUSTER,  

                            HOUSEHOLD_NUMBER) then 

                    // Households with case in household data file are either 

                    // partial or complete 

                    if ispartial(HOUSEHOLD_DICT) then 

                        inc(WS_PROGRESS_PARTIAL(i)); 

                    else 

                        inc(WS_PROGRESS_COMPLETED(i)); 

                    endif; 

                endif; 

            endif; 

        enddo; 

 

        // Remaining households are assigned households that are not complete 

        WS_PROGRESS_REMAINING(i) =  

                          WS_PROGRESS_ASSIGNED(i) - WS_PROGRESS_COMPLETED(i); 

        inc(i); 

    endfor; 

 

Now we can pass the WS_PROGRESS_REC directly to setreportdata. We will also pass the totals so that 

we can fill in the total row of the table. 

    // Set data to be replaced in report 

    setreportdata(clear);    

    setreportdata(WS_PROGRESS_REC); 

     

    numeric totalAssignedHouseholds = sum(WS_PROGRESS_ASSIGNED); 

    numeric totalCompleteHouseholds = sum(WS_PROGRESS_COMPLETED); 

    numeric totalPartialHouseholds = sum(WS_PROGRESS_PARTIAL); 

    numeric totalRemainingHousehold = sum(WS_PROGRESS_REMAINING); 

    setreportdata("assigned", maketext("%d", totalAssignedHouseholds)); 

    setreportdata("complete", maketext("%d", totalCompleteHouseholds)); 

    setreportdata("partial", maketext("%d", totalPartialHouseholds)); 

    setreportdata("remaining", maketext("%d", totalRemainingHousehold)); 

 

    // create and view the report 

    report("Reports/progress.html"); 

 

end; 

To support the progress record in our template we will use a special syntax for repeating a block of 

HTML. If we put the name of a repeating record in {{}} preceded by a # sign, that tells the system to 

repeat the HTML that follows for each occurrence of the record. To mark the end of the repeated HTML 

use the record name preceded by / in {{}}. In our report we can make a row of the table repeat once for 

each occurrence of WS_PROGRESS_REC with the following: 

                    {{#WS_PROGRESS_REC}} 

                    <tr> 

                        <td>{{WS_PROGRESS_INTERVIEWER}}</td> 

                        <td>{{WS_PROGRESS_ASSIGNED}}</td> 

                        <td>{{WS_PROGRESS_COMPLETED}}</td> 

                        <td>{{WS_PROGRESS_PARTIAL}}</td> 

                        <td>{{WS_PROGRESS_REMAINING}}</td> 

                    </tr> 

                    {{/WS_PROGRESS_REC}} 

 

Using this syntax, any names of items in the repeating record in {{}} will be replaced with the value of 

that item for each occurrence of the record. 

After adding the repeating row and the total row at the bottom of the table our table template will be: 

        <table class="table table-bordered"> 

                <thead> 

                    <tr> 

                        <th>Interviewer</th> 

                        <th>Assigned</th> 

                        <th>Complete</th> 

                        <th>Partial</th> 

                        <th>Remaining</th> 

                    </tr> 

                </thead> 

                <tbody> 

                    {{#WS_PROGRESS_REC}} 

                    <tr> 

                        <td>{{WS_PROGRESS_INTERVIEWER}}</td> 

                        <td>{{WS_PROGRESS_ASSIGNED}}</td> 

                        <td>{{WS_PROGRESS_COMPLETED}}</td> 

                        <td>{{WS_PROGRESS_PARTIAL}}</td> 

                        <td>{{WS_PROGRESS_REMAINING}}</td> 

                    </tr> 

                    {{/WS_PROGRESS_REC}} 

                </tbody> 

                <tfoot> 

                    <tr> 

                        <th>Total</th> 

                        <th>{{assigned}}</th> 

                        <th>{{complete}}</th> 

                        <th>{{partial}}</th> 

                        <th>{{remaining}}</th> 

                    </tr> 

                </tfoot> 

            </table> 

 

This will give us both the totals and the data for each interviewer. 

Group Exercise 

Implement the household status report. It will be a table that shows all the households in all clusters 

with status "household interviews". For each household, display the household number, name of the 

interviewer it is assigned to and the status (not started, partial or complete) of the household. 

 

  Household number  Assigned to  Status 

  1    Willis Catron  Not started 

  2    Zelma Hawke  Partial 

  3    Zelma Hawke  Complete 

 

Create a new record in the working storage dictionary to hold the rows of the table. Loop through the 

household assignments file and add a new occurrence to the record for each household in a cluster with 

status "Interview households". You can use the getClusterStatus() helper function to check the cluster 

status. Look up the name of the interviewer from the staff file and compute the status based on 

whether or not there is a corresponding case in the household data file and whether or not that case is 

partially saved. 


